<commit_before>{ lib, stdenv, fetchurl, pkgconfig, libpulseaudio, alsaLib, libcap
, CoreAudio, CoreServices, AudioUnit
, usePulseAudio }:

stdenv.mkDerivation rec {
  version = "1.2.0";
  name = "libao-${version}";
  src = fetchurl {
    url = "http://downloads.xiph.org/releases/ao/${name}.tar.gz";
    sha256 = "1bwwv1g9lchaq6qmhvj1pp3hnyqr64ydd4j38x94pmprs4d27b83";
  };

  outputs = [ "out" "dev" "man" "doc" ];

  buildInputs =
    [ pkgconfig ] ++
    lib.optional usePulseAudio libpulseaudio ++
    lib.optional stdenv.isLinux alsaLib ++
    lib.optional stdenv.isLinux libcap ++
    lib.optionals stdenv.isDarwin [ CoreAudio CoreServices AudioUnit ];

  meta = {
    longDescription = ''
      Libao is Xiph.org's cross-platform audio library that allows
      programs to output audio using a simple API on a wide variety of
      platforms.
    '';
    homepage = https://xiph.org/ao/;
    license = stdenv.lib.licenses.gpl2;
    maintainers = with stdenv.lib.maintainers; [ fuuzetsu ];
    platforms = with stdenv.lib.platforms; unix;
  };
}<commit_msg>libao: 1.2.0 -> 1.2.2<commit_after>{ stdenv, lib, fetchFromGitHub, autoreconfHook, pkgconfig, libpulseaudio, alsaLib, libcap
, CoreAudio, CoreServices, AudioUnit
, usePulseAudio }:

stdenv.mkDerivation rec {
  version = "1.2.2";
  name = "libao-${version}";

  # the github mirror is more up to date than downloads.xiph.org
  src = fetchFromGitHub {
    owner  = "xiph";
    repo   = "libao";
    rev    = "${version}";
    sha256 = "0svgk4sc9kdhcsfyvbvgm5vpbg3sfr6z5rliflrw49v3x2i4vxq5";
  };

  configureFlags = [
    "--disable-broken-oss"
    "--enable-alsa-mmap"
  ];

  outputs = [ "out" "dev" "man" "doc" ];

  buildInputs = [ ] ++
    lib.optional  usePulseAudio   libpulseaudio ++
    lib.optionals stdenv.isLinux  [ alsaLib libcap ] ++
    lib.optionals stdenv.isDarwin [ CoreAudio CoreServices AudioUnit ];

  nativeBuildInputs = [ autoreconfHook pkgconfig ];

  meta = with stdenv.lib; {
    longDescription = ''
      Libao is Xiph.org's cross-platform audio library that allows
      programs to output audio using a simple API on a wide variety of
      platforms.
    '';
    homepage = https://xiph.org/ao/;
    license = licenses.gpl2;
    maintainers = with maintainers; [ fuuzetsu ];
    platforms = with platforms; unix;
  };
}
{'fraction_non_alphanumeric': 0.12240484429065744, 'fraction_numerical': 0.022058823529411766, 'mean_word_length': 3.2518382352941178, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2548810', 'n_tokens_mistral': 870, 'n_tokens_neox': 828, 'n_words': 184}